

# MOVE

This chapter covers the following topics:

- Function
- Syntax 1 - MOVE ROUNDED
- Syntax 2 - MOVE SUBSTRING
- Syntax 3 - MOVE BY NAME / POSITION
- Syntax 4 - MOVE EDITED (Edit Mask Specified with operand2)
- Syntax 5 - MOVE EDITED (Edit Mask Specified with operand1)
- Syntax 6 - MOVE LEFT / RIGHT JUSTIFIED
- Syntax 7 - MOVE NORMALIZED
- Syntax 8 - MOVE ENCODED
- Examples

Related Statements: ADD | COMPRESS | COMPUTE | DIVIDE | EXAMINE | MOVE ALL | MULTIPLY | RESET | SEPARATE | SUBTRACT

Belongs to Function Group: *Arithmetic and Data Movement Operations*

---

## Function

The MOVE statement is used to move the value of an operand to one or more operands (field or array).

A MOVE statement with multiple target operands is identical to the corresponding individual MOVE statements:

```
MOVE #SOURCE TO #TARGET1 #TARGET2
```

is identical to

```
MOVE #SOURCE TO #TARGET1
MOVE #SOURCE TO #TARGET2
```

Example:

```
DEFINE DATA LOCAL
1 #ARRAY(I4/1:3) INIT <3,0,9>
1 #INDEX(I4)
1 #RESULT(I4)
END-DEFINE
*
#INDEX := 1
MOVE #ARRAY(#INDEX) TO #INDEX          /* #INDEX is 3
                        #RESULT        /* #RESULT is 9
```

\*

```
#INDEX := 2
MOVE #ARRAY(#INDEX) TO #INDEX /* #INDEX is 0
                                #ARRAY(3) /* returns run time error NAT1316
```

If *operand2* is a dynamic variable, its length may be modified by the MOVE operation. The current length of a dynamic variable can be ascertained by using the system variable \*LENGTH. For general information on the dynamic variable, see the section *Using Dynamic and Large Variables* in the *Programming Guide*.

If *operand2* is of format C, *operand1* may also be specified as parameter. Valid parameters are:

| Parameters that can be specified with the MOVE statement |                      | Specification (S = at statement level, E = at element level) |
|--|----------------------|--|
| AD   | Attribute Definition | SE   |
| CD   | Color Definition     | S  |

For more information on data transfer compatibility and the rules for data transfer, see the section *Data Transfer* in the *Programming Guide*.

## Other Considerations

If a database field is used as the result field, the MOVE operation results in an update only to the internal value of the field as used within the program. The value of the field in the database remains unchanged.

A Natural system function may be used only if the MOVE statement is specified in conjunction with an AT BREAK, AT END OF DATA or AT END OF PAGE statement.

See also the section *Rules for Arithmetic Assignment* in the *Programming Guide*.

### Note:

If *operand1* is a time variable (Format T), only the time component of the variable content is transferred, but not the date component (except with MOVE EDITED, described under *Syntax 4* and *Syntax 5*).

## Syntax 1 - MOVE ROUNDED

|   |
|---|
| <b>MOVE</b> [ROUNDED] <i>operand1</i> [(parameter)] <b>TO</b> <i>operand2</i> ... |
|---|

For an explanation of the symbols used in the syntax diagrams below, see *Syntax Symbols*.

Operand Definition Table:

| Operand         | Possible Structure |   |   |   | Possible Formats |   |   |   |   |   |   |   |   |   |   |   | Referencing Permitted | Dynamic Definition |     |
|-----------------|--------------------|---|---|---|------------------|---|---|---|---|---|---|---|---|---|---|---|-----------------------|--------------------|-----|
| <i>operand1</i> | C                  | S | A | N | A                | U | N | P | I | F | B | D | T | L | C | G | O                     | yes                | no  |
| <i>operand2</i> |                    | S | A | M | A                | U | N | P | I | F | B | D | T | L | C | G | O                     | yes                | yes |

## Syntax Element Description:

| Syntax Element   | Description  |
|------------------|--|
| MOVE<br>ROUNDED  | <p><b>MOVE ROUNDED Option:</b></p> <p>This option causes <i>operand2</i> to be rounded.</p> <p>ROUNDED is ignored if <i>operand2</i> is not numeric.</p> <p>If <i>operand2</i> is of format N or P and <i>operand2</i> is specified more than once, ROUNDED is ignored for target operands with seven positions after the decimal point.</p> <p>See also <i>Example 1 - Various Samples of MOVE Statement Usage</i>.</p>   |
| <i>parameter</i> | <p><b>Parameter:</b></p> <p>As <i>parameter</i>, you can specify the option PM=I or the session parameter DF:</p> <p><b>PM=I    Right-to-Left Display Option:</b></p> <p>In order to support languages whose writing direction is from right to left, you can specify PM=I so as to transfer the value of <i>operand1</i> in inverse (right-to-left) direction to <i>operand2</i>.</p> <p>For example, as a result of the following statements, the content of #B would be ZYX:</p> <pre>MOVE 'XYZ' TO #A MOVE #A (PM=I) TO #B</pre> <p>PM=I can only be specified if <i>operand2</i> has alphanumeric format.</p> <p>Any trailing blanks in <i>operand1</i> will be removed , then the value is reversed and moved to <i>operand2</i>. If <i>operand1</i> is not of alphanumeric format, the value will be converted to alphanumeric format before it is reversed.</p> <p>See also the use of PM=I in conjunction with MOVE LEFT/RIGHT JUSTIFIED.</p> <p><b>DF    Date Format:</b></p> <p>If <i>operand1</i> is a date variable and <i>operand2</i> is an alphanumeric field, you can specify the session parameter DF as parameter for this date variable.</p> |

## Syntax 2 - MOVE SUBSTRING

MOVE { *operand1* } [(parameter)] TO { *operand2* } ...  
 { SUBSTRING (*operand1,operand3,operand4*) } { SUBSTRING (*operand2,operand5,operand6*) }

For an explanation of the symbols used in the syntax diagrams below, see *Syntax Symbols*.

Operand Definition Table:

| Operand         | Possible Structure |   |   |  | Possible Formats |   |   |   |   |  |    |  |  |  | Referencing Permitted | Dynamic Definition |    |
|-----------------|--------------------|---|---|--|------------------|---|---|---|---|--|----|--|--|--|-----------------------|--------------------|----|
| <i>operand1</i> | C                  | S | A |  | A                | U |   |   |   |  | B  |  |  |  |                       | yes                | no |
| <i>operand2</i> |                    | S | A |  | A                | U |   |   |   |  | B  |  |  |  |                       | yes                | no |
| <i>operand3</i> | C                  | S |   |  |                  |   | N | P | I |  | B* |  |  |  |                       | yes                | no |
| <i>operand4</i> | C                  | S |   |  |                  |   | N | P | I |  | B* |  |  |  |                       | yes                | no |
| <i>operand5</i> | C                  | S |   |  |                  |   | N | P | I |  | B* |  |  |  |                       | yes                | no |
| <i>operand6</i> | C                  | S |   |  |                  |   | N | P | I |  | B* |  |  |  |                       | yes                | no |

\* See text.

Syntax Element Description:

| Syntax Element    | Description  |
|-------------------|--|
| MOVE<br>SUBSTRING | <p><b>MOVE SUBSTRING Option:</b></p> <p>Without the SUBSTRING option, the whole content of a field is moved.</p> <p>The SUBSTRING option allows you to move only a certain part of an alphanumeric, Unicode or a binary field. After the field name (<i>operand1</i>) in the SUBSTRING clause you specify first the starting position (<i>operand3</i>) and then the length (<i>operand4</i>) of the field portion to be moved.</p> <p>If the underlying field format of <i>operand1</i> is</p> <ul style="list-style-type: none"> <li>● alphanumeric (A) or binary (B), then the values supplied with <i>operand3</i> or <i>operand4</i> are considered as byte numbers;</li> <li>● Unicode (U), then the values supplied with <i>operand3</i> or <i>operand4</i> are considered as number of Unicode code units; that is, as double-bytes.</li> </ul> <p>For example, to move the 5th to 12th position inclusive of the value in a field #A into a field #B, you would specify:</p> <pre>MOVE SUBSTRING( #A, 5, 8 ) TO #B</pre> <p>If <i>operand1</i> is a dynamic variable, the specified field portion to be moved must be within its current length; otherwise, a runtime error will occur.</p> <p>Also, you can move a value of an alphanumeric, Unicode or binary field into a certain part of the target field. After the field name (<i>operand2</i>) in the SUBSTRING clause you specify first the starting position (<i>operand5</i>) and then the length (<i>operand6</i>) of the field portion into which the value is to be moved.</p> <p>If the underlying field format of <i>operand2</i> is</p> <ul style="list-style-type: none"> <li>● alphanumeric (A) or binary (B), then the values supplied with <i>operand5</i> or <i>operand6</i> are considered as byte numbers;</li> <li>● Unicode (U), then the values supplied with <i>operand3</i> or <i>operand4</i> are considered as number of Unicode code units; that is, as double-bytes.</li> </ul> <p>For example, to move the value of a field #A into the 3rd to 6th position inclusive of a field #B, you would specify:</p> <pre>MOVE #A TO SUBSTRING( #B, 3, 4 )</pre> <p>If <i>operand2</i> is a dynamic variable, the specified starting position (<i>operand5</i>) must not be greater than the variable's current length plus 1; a greater starting position will lead to a runtime error, because it would cause an undefined gap within the content of <i>operand2</i>.</p> <p>If <i>operand3/operand5</i> or <i>operand4/operand6</i> is a binary variable, it may be used only with a length of less than or equal to 4.</p> <p>If you omit <i>operand3/operand5</i>, the starting position is assumed to be 1. If you omit <i>operand4/operand6</i>, the length is assumed to range from the starting position to the end of the field.</p> <p>If <i>operand2</i> is a dynamic variable and the specified starting position (<i>operand5</i>) is the variable's current length plus 1, which means that the MOVE operation is used to increase the length of the variable, <i>operand6</i> must be specified in order to determine the new length of the variable.</p> <p><b>Note:</b><br/>MOVE with the SUBSTRING option is a byte-by-byte move (that is, the rules described under <i>Rules for Arithmetic Assignment</i> in the <i>Programming Guide</i> do not apply).</p> |

## Syntax 3 - MOVE BY NAME / POSITION

|   |
|---|
| $\text{MOVE BY } \left\{ \begin{array}{l} \text{[NAME]} \\ \text{POSITION} \end{array} \right\} \text{ operand1 TO operand2}$ |
|---|

Operand Definition Table:

| Operand         | Possible Structure | Possible Formats | Referencing Permitted | Dynamic Definition |
|-----------------|--------------------|------------------|-----------------------|--------------------|
| <i>operand1</i> | G                  |                  | yes                   | no                 |
| <i>operand2</i> | G                  |                  | yes                   | no                 |

Syntax Element Description:

| Syntax Element                                  | Description  |
|---|--|
| MOVE BY NAME <i>operand1</i> TO <i>operand2</i> | <p><b>MOVE BY NAME Option:</b></p> <p>This option is used to move individual fields contained in a data structure to another data structure, independent of their position in the structure.</p> <p>A field is moved only if its name appears in both structures (this includes REDEFINED fields as well as fields resulting from a redefinition). The individual fields may be of any format. The operands can also be views.</p> <p><b>Note:</b><br/>The sequence of the individual moves is determined by the sequence of the fields in <i>operand1</i>.</p> <p>See also <i>Example 2 - MOVE BY NAME Statement</i>.</p> <p><b>MOVE BY NAME with Arrays:</b></p> <p>If the data structures contain arrays, these will internally be assigned the index ( * ) when moved; this may lead to an error if the arrays do not comply with the rules for assignment operations with arrays; see the section <i>Processing of Arrays</i> in the <i>Programming Guide</i>.</p> <p>See also <i>Example 3 - MOVE BY NAME with Arrays</i>.</p> |

| Syntax Element   | Description  |
|--|--|
| <p>MOVE BY POSITION <i>operand1</i> TO <i>operand2</i></p> | <p><b>MOVE BY POSITION Option:</b></p> <p>This option allows you to move the contents of fields in a group to another group, regardless of the field names.</p> <p>The values are moved field by field from one group to the other in the order in which the fields are defined (this does not include fields resulting from a redefinition).</p> <p>The individual fields may be of any format. The number of fields in each group must be the same; also, the level structure and array dimensions of the fields must match. Format conversion is done according to the rules for arithmetic assignment; see the section <i>Rules for Arithmetic Assignments</i> in the <i>Programming Guide</i>. The operands can also be views.</p> <p>See also <i>Example 4 - MOVE BY POSITION</i>.</p> |

## Syntax 4 - MOVE EDITED (Edit Mask Specified with *operand2*)

|   |
|---|
| <p>MOVE EDITED <i>operand1</i> TO <i>operand2</i> { (EM=<i>value</i>) }<br/>                 { (EMU=<i>value</i>) }</p> |
|---|

Operand Definition Table:

| Operand         | Possible Structure |   |   |  | Possible Formats |   |   |   |   |   |   |   |   |   |  |  | Referencing Permitted | Dynamic Definition |     |     |
|-----------------|--------------------|---|---|--|------------------|---|---|---|---|---|---|---|---|---|--|--|-----------------------|--------------------|-----|-----|
| <i>operand1</i> | C                  | S | A |  | A                | U |   |   |   |   |   |   | B |   |  |  |                       |                    | yes | no  |
| <i>operand2</i> |                    | S | A |  | A                | U | N | P | I | F | B | D | T | L |  |  |                       |                    | yes | yes |

Syntax Element Description:

| Syntax Element | Description  |
|----------------|--|
| MOVE<br>EDITED | <p><b>MOVE EDITED Option:</b></p> <p>If an edit mask is specified for <i>operand2</i>, the value of <i>operand1</i> will be placed into <i>operand2</i> using this edit mask.</p> <p>The edit mask can be considered as an <i>input</i> edit mask for <i>operand2</i>, that is used to specify at which positions in the alphanumeric contents of <i>operand1</i> the significant input data for <i>operand2</i> can be found.</p> <p>If the edit mask refers more characters or digits than existent in <i>operand2</i>, it is truncated accordingly. The length of <i>operand1</i> may not be smaller than the length of the input value represented by the edit mask. If <i>operand1</i> is longer than the edit mask length, all the overhanging data is ignored.</p> <p>Under the pre-condition not to have an <i>operand1</i> length larger than the edit mask length, you may regard a</p> <pre>MOVE EDITED operand1 TO operand2 (EM=value)</pre> <p>operation like the execution of</p> <pre>STACK TOP DATA operand1 INPUT operand2 (EM=value)</pre> <p>See also <i>Example 1 - Various Samples of MOVE Statement Usage</i>.</p> |
| EM             | <p><b>Edit Mask:</b></p> <p>For details on edit masks, see the session parameter EM in the <i>Parameter Reference</i>.</p>   |
| EMU            | <p><b>Unicode Edit Mask:</b></p> <p>For details on Unicode edit masks, see the session parameter EMU in the <i>Parameter Reference</i>.</p>  |

## Syntax 5 - MOVE EDITED (Edit Mask Specified with *operand1*)

|   |
|---|
| <pre>MOVE EDITED operand1 { (EM=value) } TO operand2                      { (EMU=value) }</pre> |
|---|

Operand Definition Table:



| Operand         | Possible Structure |   |   |  | Possible Formats |   |   |   |   |   |   |   |   |   |   |  | Referencing Permitted | Dynamic Definition |     |
|-----------------|--------------------|---|---|--|------------------|---|---|---|---|---|---|---|---|---|---|--|-----------------------|--------------------|-----|
| <i>operand1</i> | C                  | S | A |  | N                | A | U | N | P | I | F | B | D | T | L |  |                       | yes                | no  |
| <i>operand2</i> |                    | S | A |  |                  | A | U |   |   |   |   | B |   |   |   |  |                       | yes                | yes |

Syntax Element Description:

| Syntax Element | Description  |
|----------------|--|
| MOVE<br>EDITED | <p><b>MOVE EDITED Option:</b></p> <p>If an edit mask is specified for <i>operand1</i>, the edit mask will be applied to <i>operand1</i> and the result will be moved to <i>operand2</i>.</p> <p>The edit mask can be considered as an <i>output</i> edit mask for <i>operand1</i>, that is used to create an alphanumeric string with the layout and length described by the edit mask. Besides data characters or digits originating from <i>operand1</i>, you may include additional decoration characters into the output string.</p> <p>If the edit mask refers more characters or digits than existent in <i>operand1</i>, it is truncated accordingly. The length of the created output string (resulting from <i>operand1</i> value after the edit mask has been applied) must not exceed the length of <i>operand2</i>.</p> <p>Under the pre-condition not to have an <i>operand2</i> length smaller than the edit mask length, you may regard a</p> <pre>MOVE EDITED <i>operand1</i> (EM=<i>value</i>) TO <i>operand2</i></pre> <p>operation like a</p> <pre>WRITE <i>operand1</i> (EM=<i>value</i>)</pre> <p>that does not write the output to the screen, but fills it into variable <i>operand2</i>.</p> <p>See also <i>Example 1 - Various Samples of MOVE Statement Usage</i>.</p> |
| EM             | <p><b>Edit Mask:</b></p> <p>For details on edit masks, see the session parameter EM in the <i>Parameter Reference</i>.</p>   |
| EMU            | <p><b>Unicode Edit Mask:</b></p> <p>For details on Unicode edit masks, see the session parameter EMU in the <i>Parameter Reference</i>.</p>  |

## Syntax 6 - MOVE LEFT / RIGHT JUSTIFIED

MOVE { LEFT } [JUSTIFIED] operand1 [(parameter)] TO operand2  
 { RIGHT }

Operand Definition Table:

| Operand         | Possible Structure |   |   |   | Possible Formats |   |   |   |   |   |   |   |   |   |  |  | Referencing Permitted | Dynamic Definition |
|-----------------|--------------------|---|---|---|------------------|---|---|---|---|---|---|---|---|---|--|--|-----------------------|--------------------|
| <i>operand1</i> | C                  | S | A | N | A                | U | N | P | I | F | B | D | T | L |  |  | yes                   | no                 |
| <i>operand2</i> |                    | S | A |   | A                | U |   |   |   |   |   |   |   |   |  |  | yes                   | yes                |

Syntax Element Description:

| Syntax Element                    | Description   |
|-----------------------------------|---|
| MOVE LEFT /<br>RIGHT<br>JUSTIFIED | <p><b>MOVE LEFT / RIGHT JUSTIFIED Options:</b></p> <p>This option is used to cause the values to be moved to be left- or right-justified in <i>operand2</i>.</p> <p>MOVE LEFT/RIGHT JUSTIFIED cannot be used if <i>operand2</i> is a dynamic variable.</p>  |
| MOVE LEFT<br>JUSTIFIED            | <p><b>MOVE LEFT Option:</b></p> <p>With MOVE LEFT JUSTIFIED, any leading blanks in <i>operand1</i> are removed before the value is placed left-justified into <i>operand2</i>. The remainder of <i>operand2</i> will then be filled with blanks. If the value is longer than <i>operand2</i>, the value will be truncated on the right-hand side.</p>   |
| MOVE RIGHT<br>JUSTIFIED           | <p><b>RIGHT JUSTIFIED Option:</b></p> <p>With MOVE RIGHT JUSTIFIED, any trailing blanks in <i>operand1</i> are truncated before the value is placed right-justified into <i>operand2</i>. The remainder of <i>operand2</i> will then be filled with blanks. If the value is longer than <i>operand2</i>, the value will be truncated on the left-hand side.</p> <p>See also <i>Example 1 - Various Samples of MOVE Statement Usage</i>.</p>   |
| <i>parameter</i>                  | <p><b>Parameter:</b></p> <p>When you use MOVE LEFT/RIGHT JUSTIFIED in conjunction with PM=I, the move is performed in the following steps:</p> <ol style="list-style-type: none"> <li>1. If <i>operand1</i> is not of alphanumeric format, the value is converted to alphanumeric format.</li> <li>2. Any trailing blanks in <i>operand1</i> are removed.</li> <li>3. In the case of LEFT JUSTIFIED, any leading blanks in <i>operand1</i> are also removed.</li> <li>4. The value is reversed, and then moved to <i>operand2</i>.</li> <li>5. If applicable, the remainder of <i>operand2</i> is filled with blanks, or the value is truncated (see above).</li> </ol> |

## Syntax 7 - MOVE NORMALIZED

The MOVE NORMALIZED statement converts a Unicode string into the "Unicode Normalization Form C" (NFC). The resulting Unicode string does no longer contain combining sequences for characters which are available as pre-composed characters.

If the format of the target operand is not Unicode itself, an implicit conversion from Unicode into the target operand takes place - during this conversion the default code page (see system variable \*CODEPAGE) will be used.





| Syntax Element              | Description   |
|-----------------------------|---|
| MOVE ENCODED                | <p><b>MOVE ENCODED Option:</b></p> <p>This option converts a character string, encoded in one code page, into the equivalent character string of another code page.</p> <p><b>Note:</b><br/>Natural uses the International Components for Unicode (ICU) library for Unicode conversion. For more information, see <a href="http://icu.sourceforge.net/userguide/">http://icu.sourceforge.net/userguide/</a>.</p>  |
| <i>operand1</i>             | <p><b>Source Operand:</b></p> <p><i>operand1</i> contains the string to be converted.</p>   |
| CODEPAGE<br><i>operand2</i> | <p><b>Code Page of Source Operand:</b></p> <p>As <i>operand2</i>, you specify the code page of <i>operand1</i>.</p> <p>Can only be supplied if <i>operand1</i> is of format A or B. See Note 1.</p>   |
| TO <i>operand3</i>          | <p><b>Target Operand:</b></p> <p><i>operand3</i> receives the converted string.</p> <p>If the conversion result does not fit into the target field, the result is padded or truncated, respectively, and as padding character the blank of the resulting code page is used.</p> <p>If the target field is defined as a dynamic variable, no padding or truncation is needed, since the length of the dynamic variable is automatically adjusted to the length of the conversion result.</p> |
| CODEPAGE<br><i>operand4</i> | <p><b>Code Page of Target Operand:</b></p> <p>As <i>operand4</i>, you specify the code page of <i>operand3</i>.</p> <p>Can only be supplied if <i>operand3</i> is of format A or B. See Note 1.</p>   |
| GIVING<br><i>operand5</i>   | <p><b>GIVING Clause:</b></p> <p>If you omit this clause, a Natural error message is returned if an error occurs.</p> <p>If you specify the keyword GIVING, <i>operand5</i> returns 0 or the Natural error code instead of the Natural error message.</p> <p>If the target gets truncated, no Natural error message is given, but when the keyword GIVING is used, <i>operand5</i> will contain an appropriate error code to indicate truncation.</p>  |

**Notes:**

1. If a code page operand is not supplied, then the default code page (value of the system variable \*CODEPAGE) is used.
2. If the session parameter CPCVERR in the statement SET GLOBALS or in the system command GLOBALS is set to ON, an error is output if at least one character of the source field could not be

converted properly into the destination code page, but was replaced in the target field by a substitution character.

### Examples of MOVE ENCODED:

```
MOVE ENCODED A-FIELD1 TO A-FIELD2
```

Invalid: This results in a syntax error, since the code page names are taken by default and are the same for *operand1* and *operand3*.

```
MOVE ENCODED A-FIELD1 CODEPAGE 'IBM01140' TO A-FIELD2 CODEPAGE 'IBM01140'
```

Invalid: This results in an error, since the coded code page names are the same for *operand1* and *operand3*.

```
MOVE ENCODED A-FIELD1 CODEPAGE 'IBM01140' TO A-FIELD2 CODEPAGE 'IBM037'
```

Valid: The string in A-FIELD1 which is coded in IBM01140 is converted into A-FIELD2 which is coded in IBM037.

```
MOVE ENCODED U-FIELD TO U-FIELD
```

Invalid: This results in an error, since at least one operand must be of format A or B.

```
MOVE ENCODED U-FIELD TO A-FIELD
```

Valid: The Unicode string in U-FIELD which, considered to be encoded in UTF-16, is converted into the alphanumeric A-FIELD in the default code page (\*CODEPAGE).

```
MOVE ENCODED A-FIELD TO U-FIELD
```

Valid: The string in A-FIELD which, considered to be encoded in the default code page (\*CODEPAGE), is converted into the Unicode field U-FIELD.

```
MOVE ENCODED A100-FIELD CODEPAGE 'IBM1140' TO A50-FIELD CODEPAGE 'IBM037'
```

Valid: Conversion is done from A100-FIELD (format/length: A100) to A50-FIELD (format/length: A50), using the relevant code pages. The target is truncated. No Natural error message is returned.

```
MOVE ENCODED A100-FIELD CODEPAGE 'IBM1140' TO A50-FIELD CODEPAGE 'IBM037' GIVING RC-FIELD
```

Valid: Conversion is done from A100-FIELD (format/length: A100) to A50-FIELD (format/length: A50), using the relevant code pages. The target is truncated. Since a GIVING clause is supplied, the RC-FIELD receives an error code, indicating that a value truncation has taken place.

## Examples

- Example 1 - Various Samples of MOVE Statement Usage
- Example 2 - MOVE BY NAME
- Example 3 - MOVE BY NAME with Arrays

- Example 4- MOVE BY POSITION

## Example 1 - Various Samples of MOVE Statement Usage

```

** Example 'MOVEX1': MOVE
*****
DEFINE DATA LOCAL
1 #A (N3)
1 #B (A5)
1 #C (A2)
1 #D (A7)
1 #E (N1.0)
1 #F (A5)
1 #G (N3.2)
1 #H (A6)
END-DEFINE
*
MOVE 5 TO #A
WRITE NOTITLE 'MOVE 5 TO #A'          30X '=' #A
*
MOVE 'ABCDE' TO #B #C #D
WRITE 'MOVE ABCDE TO #B #C #D'      20X '=' #B '=' #C '=' #D
*
MOVE -1 TO #E
WRITE 'MOVE -1 TO #E'                28X '=' #E
*
MOVE ROUNDED 1.995 TO #E
WRITE 'MOVE ROUNDED 1.995 TO #E'    18X '=' #E
*
*
MOVE RIGHT JUSTIFIED 'ABC' TO #F
WRITE 'MOVE RIGHT JUSTIFIED 'ABC'' TO #F'      10X '=' #F
*
MOVE EDITED '003.45' TO #G (EM=999.99)
WRITE 'MOVE EDITED ''003.45'' TO #G (EM=999.99)'  4X '=' #G
*
MOVE EDITED 123.45 (EM=999.99) TO #H
WRITE 'MOVE EDITED 123.45 (EM=999.99) TO #H'    6X '=' #H
*
END

```

### Output of Program MOVEX1:

```

MOVE 5 TO #A          #A:      5
MOVE ABCDE TO #B #C #D  #B: ABCDE #C: AB #D: ABCDE
MOVE -1 TO #E         #E:     -1
MOVE ROUNDED 1.995 TO #E  #E:      2
MOVE RIGHT JUSTIFIED 'ABC' TO #F  #F:   ABC
MOVE EDITED '003.45' TO #G (EM=999.99)  #G:    3.45
MOVE EDITED 123.45 (EM=999.99) TO #H  #H:  123.45

```

## Example 2 - MOVE BY NAME

```

** Example 'MOVEX2': MOVE BY NAME
*****
DEFINE DATA LOCAL
1 #SBLOCK
  2 #FIELD A (A10) INIT <'AAAAAAAAAA'>
  2 #FIELD B (A10) INIT <'BBBBBBBBBB'>
  2 #FIELD C (A10) INIT <'CCCCCCCCCC'>
  2 #FIELD D (A10) INIT <'DDDDDDDDDD'>

```



```

1 #TBLOCK
  2 #FIELD1 (A15) INIT <' '>
  2 #FIELD2 (A10) INIT <' '>
  2 #FIELD3 (A20) INIT <' '>
  2 #FIELD4 (A10) INIT <' '>
  2 #FIELD5 (A10) INIT <' '>
  2 #FIELD6 (A10) INIT <' '>
END-DEFINE
*
MOVE BY NAME #SBLOCK TO #TBLOCK
*
WRITE NOTITLE 'CONTENTS OF #TBLOCK AFTER MOVE BY NAME:'
  // '=' #TBLOCK.#FIELD1
  / '=' #TBLOCK.#FIELD2
  / '=' #TBLOCK.#FIELD3
  / '=' #TBLOCK.#FIELD4
  / '=' #TBLOCK.#FIELD5
  / '=' #TBLOCK.#FIELD6
*
END

```

### Contents of #TBLOCK after MOVE BY NAME Processing:

CONTENTS OF #TBLOCK AFTER MOVE BY NAME:

```

#FIELD1:
#FIELD2: AAAAAAAAAA
#FIELD3:
#FIELD4: BBBBBBBBBB
#FIELD5:
#FIELD6: CCCCCCCCCC

```

### Example 3 - MOVE BY NAME with Arrays

```

DEFINE DATA LOCAL
  1 #GROUP1
    2 #FIELD (A10/1:10)
  1 #GROUP2
    2 #FIELD (A10/1:10)
END-DEFINE
...
MOVE BY NAME #GROUP1 TO #GROUP2
...

```

In this example, the MOVE statement would internally be resolved as:

```
MOVE #GROUP1.#FIELD (*) TO #GROUP2.#FIELD (*)
```

If part of an indexed group is moved to another part of the same group, this may lead to unexpected results as shown in the example below.

```

DEFINE DATA LOCAL
  1 #GROUP1 (1:5)
    2 #FIELD1 (N1) INIT <1,2,3,4,5>
    2 REDEFINE #FIELD1
      3 #FIELD2 (N1)
END-DEFINE
...
MOVE BY NAME #GROUP1 (2:4) TO #GROUP1 (1:3)
...

```

In this example, the MOVE statement would internally be resolved as:

```
MOVE #FIELDA (2:4) TO #FIELDA (1:3) MOVE #FIELDB (2:4) TO #FIELDB (1:3)
```

First, the contents of the occurrences 2 to 4 of #FIELDA are moved to the occurrences 1 to 3 of #FIELDA; that is, the occurrences receive the following values:

|               |    |    |    |    |    |
|---------------|----|----|----|----|----|
| Occurrence:   | 1. | 2. | 3. | 4. | 5. |
| Value before: | 1  | 2  | 3  | 4  | 5  |
| Value after:  | 2  | 3  | 4  | 4  | 5  |

Then the contents of the occurrences 2 to 4 of #FIELDB are moved to the occurrences 1 to 3 of #FIELDB; that is, the occurrences receive the following values:

|               |    |    |    |    |    |
|---------------|----|----|----|----|----|
| Occurrence:   | 1. | 2. | 3. | 4. | 5. |
| Value before: | 2  | 3  | 4  | 4  | 5  |
| Value after:  | 3  | 4  | 4  | 4  | 5  |

## Example 4- MOVE BY POSITION

```
DEFINE DATA LOCAL
  1 #GROUP1
    2 #FIELD1A (N5)
    2 #FIELD1B (A3/1:3)
    2 REDEFINE #FIELD1B
      3 #FIELD1BR (A9)
  1 #GROUP2
    2 #FIELD2A (N5)
    2 #FIELD2B (A3/1:3)
    2 REDEFINE #FIELD2B
      3 #FIELD2BR (A9)
END-DEFINE
...
MOVE BY POSITION #GROUP1 TO #GROUP2
...
```

In this example, the content of #FIELD1A is moved to #FIELD2A, and the content of #FIELD1B to #FIELD2B; the fields #FIELD1BR and #FIELD2BR are not affected.